# Rules for integrands of the form $(d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p$

**1:** $\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n]) \, dx$ when $q \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{Log}[c x^n]) == \frac{b n}{x}$

Rule: If $q \in \mathbb{Z}^+$, let $u \to \int (d + e x^r)^q \, dx$, then

$$\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n]) \, dx \longrightarrow u (a + b \operatorname{Log}[c x^n]) - b n \int \frac{u}{x} \, dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^r)^q,x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[q,0]
```

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^r)^q,x]},
  Dist[(a+b*Log[c*x^n]),u] - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[q,0]
```

**2:** $\int (d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $r\,(q+1)+1=0$

Derivation: Integration by parts

Basis: If $r\,(q+1)+1=0$, then $(d+e\,x^r)^q=\partial_x\frac{x\,(d+e\,x^r)^{q+1}}{d}$

Rule: If $r\,(q+1)+1=0$, then

$$\int (d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\,\longrightarrow\,\frac{x\,(d+e\,x^r)^{q+1}\,(a+b\,\text{Log}[c\,x^n])}{d}-\frac{b\,n}{d}\int (d+e\,x^r)^{q+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  x*(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])/d - b*n/d*Int[(d+e*x^r)^(q+1),x] /;
FreeQ[{a,b,c,d,e,n,q,r},x] && EqQ[r*(q+1)+1,0]
```

**x:** $\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{d+e\,x^r}\,dx$ when $p\in\mathbb{Z}^+\land r\in\mathbb{Z}^+$

- Derivation: Algebraic expansion

- Basis: $\frac{x^m}{d+e\,x^r}==\frac{x^{m-r}}{e}-\frac{d\,x^{m-r}}{e\,(d+e\,x^r)}$

- Note: This rule produces antiderivatives in terms of $\text{PolyLog}\left[k,\,-\frac{d}{e\,x^r}\right]$

- Rule: If $p\in\mathbb{Z}^+\land r\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{d+e\,x^r}\,dx\,\longrightarrow\,\frac{1}{e}\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{x^r}\,dx-\frac{d}{e}\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{x^r\,(d+e\,x^r)}\,dx$$

- Program code:

```
(* Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_^r_.),x_Symbol] :=
  1/e*Int[(a+b*Log[c*x^n])^p/x^r,x] -
  d/e*Int[(a+b*Log[c*x^n])^p/(x^r*(d+e*x^r)),x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0] && IGtQ[r,0] *)
```

3. $\int (d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$

1. $\int (d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $p>0$

1. $\int \frac{(a+b\,\mathrm{Log}[c\,x^n])^p}{d+e\,x}\,dx$ when $p\in\mathbb{Z}^+$

1. $\int \frac{a+b\,\mathrm{Log}[c\,x]}{d+e\,x}\,dx$ when $-\frac{c\,d}{e}>0$

1: $\int \frac{\mathrm{Log}[c\,x]}{d+e\,x}\,dx$ when $e+c\,d==0$

Rule: If $e+c\,d==0$, then

$$\int \frac{\mathrm{Log}[c\,x]}{d+e\,x}\,dx \longrightarrow -\frac{1}{e}\,\mathrm{PolyLog}[2,\,1-c\,x]$$

Program code:

```
Int[Log[c_.*x_]/(d_+e_.*x_),x_Symbol] :=
  -1/e*PolyLog[2,1-c*x] /;
FreeQ[{c,d,e},x] && EqQ[e+c*d,0]
```

2: $\int \frac{a+b\,\mathrm{Log}[c\,x]}{d+e\,x}\,dx$ when $-\frac{c\,d}{e}>0$

Derivation: Algebraic expansion

Basis: If $-\frac{c\,d}{e}>0$, then $\mathrm{Log}[c\,x]==\mathrm{Log}\left[-\frac{c\,d}{e}\right]+\mathrm{Log}\left[-\frac{e\,x}{d}\right]$

– Note: Resulting integrand is of the form required by the above rule.

– Rule: If $-\frac{c\,d}{e}>0$, then

$$\int \frac{a + b \operatorname{Log}[c\ x]}{d + e\ x}\,dx \longrightarrow \frac{\left(a + b \operatorname{Log}\left[-\frac{c\,d}{e}\right]\right) \operatorname{Log}[d + e\ x]}{e} + b \int \frac{\operatorname{Log}\left[-\frac{e\,x}{d}\right]}{d + e\ x}\,dx$$

## Program code:

```
Int[(a_.+b_.*Log[c_.*x_])/(d_+e_.*x_),x_Symbol] :=
  (a+b*Log[-c*d/e])*Log[d+e*x]/e + b*Int[Log[-e*x/d]/(d+e*x),x] /;
FreeQ[{a,b,c,d,e},x] && GtQ[-c*d/e,0]
```

**2:** $\int \frac{(a + b \operatorname{Log}[c\ x^n])^p}{d + e\ x}\,dx$ when $p \in \mathbb{Z}^+$

## Derivation: Integration by parts

## Basis: $\frac{1}{d+e\,x} = \frac{1}{e}\,\partial_x \operatorname{Log}\left[1 + \frac{e\,x}{d}\right]$

## Basis: $\partial_x\,(a + b \operatorname{Log}[c\ x^n])^p = \frac{b\,n\,p\,(a+b \operatorname{Log}[c\ x^n])^{p-1}}{x}$

## Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{Log}[c\ x^n])^p}{d + e\ x}\,dx \longrightarrow \frac{\operatorname{Log}\left[1 + \frac{e\,x}{d}\right]\,(a + b \operatorname{Log}[c\ x^n])^p}{e} - \frac{b\,n\,p}{e} \int \frac{\operatorname{Log}\left[1 + \frac{e\,x}{d}\right]\,(a + b \operatorname{Log}[c\ x^n])^{p-1}}{x}\,dx$$

## Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_),x_Symbol] :=
  Log[1+e*x/d]*(a+b*Log[c*x^n])^p/e - b*n*p/e*Int[Log[1+e*x/d]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0]
```

**2:** $\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{(d+e\,x)^2}\,dx$ when $p>0$

## Derivation: Integration by parts

Basis: $\frac{1}{(d+e\,x)^2} == \partial_x \frac{x}{d\,(d+e\,x)}$

Basis: $\partial_x (a+b\,\text{Log}[c\,x^n])^p == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}$

Rule: If $p>0$, then

$$\int \frac{(a+b\,\text{Log}[c\,x^n])^p}{(d+e\,x)^2}\,dx \longrightarrow \frac{x\,(a+b\,\text{Log}[c\,x^n])^p}{d\,(d+e\,x)} - \frac{b\,n\,p}{d}\int \frac{(a+b\,\text{Log}[c\,x^n])^{p-1}}{d+e\,x}\,dx$$

## Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_)^2,x_Symbol] :=
  x*(a+b*Log[c*x^n])^p/(d*(d+e*x)) - b*n*p/d*Int[(a+b*Log[c*x^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && GtQ[p,0]
```

**3:** $\int (d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$ when $p>0\,\land\,q\neq -1$

## Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

## Derivation: Integration by parts

## Basis: $\partial_x\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p == \frac{b\,n\,p\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p-1}}{x}$

## Rule: If $p>0\,\land\,q\neq -1$, then

$$\int (d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx \longrightarrow \frac{(d+e\,x)^{q+1}\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{e\,(q+1)}-\frac{b\,n\,p}{e\,(q+1)}\int\frac{(d+e\,x)^{q+1}\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p-1}}{x}\,dx$$

## Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/(e*(q+1)) - b*n*p/(e*(q+1))*Int[((d+e*x)^(q+1)*(a+b*Log[c*x^n])^(p-1))/x,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && GtQ[p,0] && NeQ[q,-1] && (EqQ[p,1] || IntegersQ[2*p,2*q] && Not[IGtQ[q,0]] || EqQ[p,2] && NeQ[q,1])
```

**2:** $\int (d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $p<-1\,\land\,q>0$

## Rule: If $p<-1\,\land\,q>0$, then

$$\int (d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx\;\longrightarrow$$

$$\frac{x\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^{p+1}}{b\,n\,(p+1)}+\frac{d\,q}{b\,n\,(p+1)}\int (d+e\,x)^{q-1}\,(a+b\,\text{Log}[c\,x^n])^{p+1}\,dx-\frac{q+1}{b\,n\,(p+1)}\int (d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^{p+1}\,dx$$

## Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x*(d+e*x)^q*(a+b*Log[c*x^n])^(p+1)/(b*n*(p+1)) +
  d*q/(b*n*(p+1))*Int[(d+e*x)^(q-1)*(a+b*Log[c*x^n])^(p+1),x] -
  (q+1)/(b*n*(p+1))*Int[(d+e*x)^q*(a+b*Log[c*x^n])^(p+1),x] /;
FreeQ[{a,b,c,d,e,n},x] && LtQ[p,-1] && GtQ[q,0]
```

4. $\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx$

1: $\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx$ when $q>0$

## Rule: If $q>0$, then

$$\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx \to \frac{x (d+e x^2)^q (a+b \operatorname{Log}[c x^n])}{2q+1} - \frac{b n}{2q+1} \int (d+e x^2)^q \, dx + \frac{2 d q}{2q+1} \int (d+e x^2)^{q-1} (a+b \operatorname{Log}[c x^n]) \, dx$$

## Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  x*(d+e*x^2)^q*(a+b*Log[c*x^n])/(2*q+1) -
  b*n/(2*q+1)*Int[(d+e*x^2)^q,x] +
  2*d*q/(2*q+1)*Int[(d+e*x^2)^(q-1)*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,n},x] && GtQ[q,0]
```

2. $\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx$ when $q<-1$

1: $\int \frac{a+b \operatorname{Log}[c x^n]}{(d+e x^2)^{3/2}} \, dx$

## Rule:

$$\int \frac{a+b \operatorname{Log}[c x^n]}{(d+e x^2)^{3/2}} \, dx \to \frac{x (a+b \operatorname{Log}[c x^n])}{d \sqrt{d+e x^2}} - \frac{b n}{d} \int \frac{1}{\sqrt{d+e x^2}} \, dx$$

## Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*Log[c*x^n])/(d*Sqrt[d+e*x^2]) - b*n/d*Int[1/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,n},x]
```

**2:** $\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx$ when $q < -1$

## Rule: If $q < -1$, then

$$\int (d+e x^2)^q (a+b \operatorname{Log}[c x^n]) \, dx \longrightarrow -\frac{x (d+e x^2)^{q+1} (a+b \operatorname{Log}[c x^n])}{2 d (q+1)} + \frac{b n}{2 d (q+1)} \int (d+e x^2)^{q+1} \, dx + \frac{2 q+3}{2 d (q+1)} \int (d+e x^2)^{q+1} (a+b \operatorname{Log}[c x^n]) \, dx$$

## Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  -x*(d+e*x^2)^(q+1)*(a+b*Log[c*x^n])/(2*d*(q+1)) +
  b*n/(2*d*(q+1))*Int[(d+e*x^2)^(q+1),x] +
  (2*q+3)/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,n},x] && LtQ[q,-1]
```

**3:** $\int \frac{a + b \operatorname{Log}[c\, x^n]}{d + e\, x^2}\, dx$

- Derivation: Integration by parts

- Basis: $\partial_x (a + b \operatorname{Log}[c\, x^n]) == \frac{b\, n}{x}$

- Rule: Let $u \to \int \frac{1}{d+e\, x^2}\, dx$, then

$$\int \frac{a + b \operatorname{Log}[c\, x^n]}{d + e\, x^2}\, dx \longrightarrow u \left(a + b \operatorname{Log}[c\, x^n]\right) - b\, n \int \frac{u}{x}\, dx$$

- Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/(d_+e_.*x_^2),x_Symbol] :=
  With[{u=IntHide[1/(d+e*x^2),x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[u/x,x]] /;
FreeQ[{a,b,c,d,e,n},x]
```

**4.** $\int \frac{a + b \operatorname{Log}[c\, x^n]}{\sqrt{d + e\, x^2}}\, dx$

**1.** $\int \frac{a + b \operatorname{Log}[c\, x^n]}{\sqrt{d + e\, x^2}}\, dx$ when $d > 0$

**1:** $\int \frac{a + b \operatorname{Log}[c\, x^n]}{\sqrt{d + e\, x^2}}\, dx$ when $d > 0 \land e > 0$

Derivation: Integration by parts

Basis: If $d > 0$, then $\frac{1}{\sqrt{d+e\, x^2}} == \partial_x \frac{\operatorname{ArcSinh}\left[\frac{\sqrt{e}\, x}{\sqrt{d}}\right]}{\sqrt{e}}$

Rule: If $d > 0 \land e > 0$, then

$$\int \frac{a + b \operatorname{Log}[c\, x^n]}{\sqrt{d + e\, x^2}}\, dx \longrightarrow \frac{\operatorname{ArcSinh}\left[\frac{\sqrt{e}\, x}{\sqrt{d}}\right] \left(a + b \operatorname{Log}[c\, x^n]\right)}{\sqrt{e}} - \frac{b\, n}{\sqrt{e}} \int \frac{\operatorname{ArcSinh}\left[\frac{\sqrt{e}\, x}{\sqrt{d}}\right]}{x}\, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  ArcSinh[Rt[e,2]*x/Sqrt[d]]*(a+b*Log[c*x^n])/Rt[e,2] - b*n/Rt[e,2]*Int[ArcSinh[Rt[e,2]*x/Sqrt[d]]/x,x] /;
FreeQ[{a,b,c,d,e,n},x] && GtQ[d,0] && PosQ[e]
```

**2:** $\int \frac{a + b \text{Log}[c x^n]}{\sqrt{d + e x^2}} \, dx$ when $d > 0 \land e \not> 0$

Derivation: Integration by parts

Basis: If $d > 0$, then $\frac{1}{\sqrt{d+e x^2}} == \partial_x \frac{\text{ArcSin}\left[\frac{\sqrt{-e} x}{\sqrt{d}}\right]}{\sqrt{-e}}$

Rule: If $d > 0 \land e \not> 0$, then

$$\int \frac{a + b \text{Log}[c x^n]}{\sqrt{d + e x^2}} \, dx \rightarrow \frac{\text{ArcSin}\left[\frac{\sqrt{-e} x}{\sqrt{d}}\right] (a + b \text{Log}[c x^n])}{\sqrt{-e}} - \frac{b n}{\sqrt{-e}} \int \frac{\text{ArcSin}\left[\frac{\sqrt{-e} x}{\sqrt{d}}\right]}{x} \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  ArcSin[Rt[-e,2]*x/Sqrt[d]]*(a+b*Log[c*x^n])/Rt[-e,2] - b*n/Rt[-e,2]*Int[ArcSin[Rt[-e,2]*x/Sqrt[d]]/x,x] /;
FreeQ[{a,b,c,d,e,n},x] && GtQ[d,0] && NegQ[e]
```

**2:** $\int \frac{a + b \text{Log}[c x^n]}{\sqrt{d + e x^2}} \, dx$ when $d \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{e}{d} x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $d \not> 0$, then

$$\int \frac{a + b \operatorname{Log}\left[c\, x^n\right]}{\sqrt{d + e\, x^2}} \, dx \longrightarrow \frac{\sqrt{1 + \frac{e}{d} x^2}}{\sqrt{d + e\, x^2}} \int \frac{a + b \operatorname{Log}\left[c\, x^n\right]}{\sqrt{1 + \frac{e}{d} x^2}} \, dx$$

–

## Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+e/d*x^2]/Sqrt[d+e*x^2]*Int[(a+b*Log[c*x^n])/Sqrt[1+e/d*x^2],x] /;
FreeQ[{a,b,c,d,e,n},x] && Not[GtQ[d,0]]
```

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Sqrt[1+e1*e2/(d1*d2)*x^2]/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*Int[(a+b*Log[c*x^n])/Sqrt[1+e1*e2/(d1*d2)*x^2],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[d2*e1+d1*e2,0]
```

**5:** $\int (d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $2\,q\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Note: If $q-\frac{1}{2}\in\mathbb{Z}$, then the terms of $\int (d+e\,x)^q\,dx$ will be algebraic functions or constants times an inverse function.

Rule: If $2\,q\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$, let $u\to\int (d+e\,x^r)^q\,dx$, then

$$\int (d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\,\longrightarrow\,u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^r)^q,x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[SimplifyIntegrand[u/x,x],x] /;
 EqQ[r,1] && IntegerQ[q-1/2] || EqQ[r,2] && EqQ[q,-1] || InverseFunctionFreeQ[u,x]] /;
FreeQ[{a,b,c,d,e,n,q,r},x] && IntegerQ[2*q] && IntegerQ[r]
```

**6:** $\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx$ when $q \in \mathbb{Z} \land (q > 0 \lor p \in \mathbb{Z}^+ \land r \in \mathbb{Z})$

- Derivation: Algebraic expansion

- Rule: If $q \in \mathbb{Z} \land (q > 0 \lor p \in \mathbb{Z}^+ \land r \in \mathbb{Z})$, then

$$\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \int (a + b \operatorname{Log}[c x^n])^p \, \text{ExpandIntegrand}[(d + e x^r)^q, x] \, dx$$

- Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n])^p,(d+e*x^r)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n,p,q,r},x] && IntegerQ[q] && (GtQ[q,0] || IGtQ[p,0] && IntegerQ[r])
```

**U:** $\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx$

- Rule:

$$\int (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \int (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx$$

- Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q,r},x]
```

**N:** $\int u^q \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^p \, dx$ when $u == d + e\, x^r$

Derivation: Algebraic normalization

– Rule: If $u == d + e\, x^r$, then

$$\int u^q \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^p \, dx \longrightarrow \int (d + e\, x)^q \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^p \, dx$$

– Program code:

```
Int[u_^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,n,p,q},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```